
Korutyny i wπtki
Odraczanie celu

I Cel X = 2, X > 1 koÒczy siÍ powodzeniem.
I Cel X > 1, X = 2 koÒczy siÍ b≥Ídem.
I Sprawdzenie warunku X > 1 naleøy odroczyÊ do czasu gdy

zmienna X przyjmie wartoúÊ.
I DziÍki odraczaniu Prolog moøe byÊ znowu bardziej

deklaratywny (koniunkcja jest przemienna).

Korutyny i wπtki
Predykat freeze/2

I Meta-predykat freeze(Var, Goal) odracza sprawdzenie
celu Goal do chwili gdy zmienna Var przyjmie wartoúÊ.

I Cel sprawdzany jest natychmiast po tym jak zmienna przyjmie
wartoúÊ.

I Niepowodzenie celu powoduje natychmiastowe wycofanie siÍ.



Korutyny i wπtki
Predykat freeze/2

?- freeze(X, X > 1), X = 2.
X = 2.

?- freeze(X, writeln(x=X)), freeze(Y, writeln(y=Y)),
f(X, Y) = f(a, b).

x=a
y=b
X = a,
Y = b.

?- freeze(X, writeln(x=X)), freeze(Y, writeln(y=Y)),
f(Y, X) = f(a, b).

y=a
x=b
X = b,
Y = a.

Korutyny i wπtki
Predykat freeze/2

?- freeze(X, (writeln(x=X), Z = c)),
freeze(Y, writeln(y=Y)),
freeze(Z, writeln(z=Z)),
f(X, Y) = f(a, b).

x=a
z=c
y=b
X = a,
Z = c,
Y = b.

Predykat freeze/2

?- X \= Y, X = a, Y = b.
false.

?- freeze(X, freeze(Y, X \= Y)), X = a, Y = b.
X = a,
Y = b.

?- freeze(X, freeze(Y, X \= Y)), X = a.
X = a,
freeze(Y, a\=Y).

?- freeze(X, freeze(Y, X \= Y)), Y = b.
Y = b,
freeze(X, freeze(b, X\=b)).

Korutyny i wπtki
Predykat when/2

I Meta-predykat when(Condition, Goal) odracza
sprawdzenie celu Goal do momentu gdy warunek Condition
bÍdzie spe≥niony.

I Warunek Condition moøe mieÊ nastÍpujπcπ postaÊ: (X ?=
Y), nonvar(X), ground(X), (Cond1, Cond2) lub (Cond1;
Cond2).

I freeze(X, G) jest równowaøne when(nonvar(X), G) ale
nie jest równowaøne when(ground(X), G).



Korutyny i wπtki
Predykat when/2

?- freeze(X, writeln(x=X)), X = f(Y), Y = a.
x=f(_G1261)
X = f(a),
Y = a.

?- when(ground(X), writeln(x=X)), X = f(Y), Y = a.
x=f(a)
X = f(a),
Y = a.

Predykat when/2

?- when((nonvar(X), nonvar(Y)), writeln(f(X, Y))),
X = a, Y = b.

f(a,b)
X = a,
Y = b.

?- when((nonvar(X); nonvar(Y)), writeln(f(X, Y))),
X = a, Y = b.

f(a,_G1523)
X = a,
Y = b.

Korutyny i wπtki
Predykat dif/2

I Predykat dif(Term1, Term2) narzuca ograniczenie, øe
termy Term1 i Term2 sπ róønymi termami.

I Jeúli Term1 i Term2 nie sπ unifikowalne, to dif(Term1,
Term2) jest natychmiast spe≥niony.

I Jeúli Term1 i Term2 sπ identyczne, to dif(Term1, Term2)
natychmiast zawodzi.

I Jeúli Term1 i Term2 mogπ zunifikowaÊ siÍ, wówczas
dif(Term1, Term2) odracza warunki zapewniajπce róønoúÊ
Term1 i Term2.

Predykat dif/2

?- dif(f(X, Y), f(X, Y)).
false.

?- dif(f(X, a), f(b, b)).
true.

?- dif(f(X, Y), f(Y, X)).
dif(Y, X).

?- dif(f(X, a), f(b, Y)).
dif(f(X, Y), f(b, a)).

Predykat dif/2

?- dif(f(X, a), f(b, Y)), X = a.
X = a.

?- dif(f(X, a), f(b, Y)), X = b.
X = b,
dif(f(b, Y), f(b, a)).

?- dif(f(X, a), f(b, Y)), X = b, Y = a.
false.

Korutyny i wπtki
Listy otwarte

I Lista otwarta reprezentuje strumieÒ termów, przy czym w
strumieniu tym mogπ pojawiaÊ siÍ kolejne termy.

I Pusty strumieÒ reprezentowany jest nieukonkretnionπ zmiennπ
_.

I Kolejne termy wpisywane sπ do nieukonkretnionego ogona
otwartej listy.

I Pusty strumieÒ termów S spe≥nia warunek warunek var(S).
I Aby zakoÒczyÊ strumieÒ termów S naleøy zunifikowaÊ go z

listπ pustπ [ ].
I W wyniku unifikacji niepustego strumienia S ze wzorcem [H |

T] zostaje pod H podstawiony pierwszy element strumienia a
T unifikuje siÍ ze strumieniem kolejnych termów.

Listy otwarte

?- L1 = [a | L2], L2 = [b | L3], L3 = [c | L0].
L1 = [a, b, c|L0],
L2 = [b, c|L0],
L3 = [c|L0].

?- L1 = [a | L2], L2 = [b | L3], L3 = [c | L0], L0 = [].
L1 = [a, b, c],
L2 = [b, c],
L3 = [c],
L0 = [].

Korutyny i wπtki
Korutyny przekazujπce sobie dane w strumieniach

Example (Podwajanie)

1, 2, 3, 4, 5 2, 4, 6, 8, 10generowanie podwajanie drukuj

2
4
6
8
10

Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

generowanie(I, J, S) :-
( I =< J
-> S = [I | T],

I1 is I+1,
generowanie(I1, J, T)

; S = []).

Korutyny i wπtki
Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

podwajanie(S1, S2) :-
freeze(S1,

( S1 = [H1 | T1]
-> H2 is 2*H1,

S2 = [H2 | T2],
podwajanie(T1, T2)

; S2 = [])).

Korutyny i wπtki
Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

drukuj(S) :-
freeze(S,

( S = [H | T]
-> writeln(H),

drukuj(T)
; true)).

Korutyny i wπtki
Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

?- drukuj(S2), podwajanie(S1, S2), generowanie(1, 5, S1).
2
4
6
8
10
S2 = [2, 4, 6, 8, 10],
S1 = [1, 2, 3, 4, 5].

Korutyny przekazujπce sobie dane w strumieniach

Example (Sito Eratostenesa)

sito

czytaj pisz

filtruj sito

2, 3, 4, 5, 6, 7, 8, 9, 10, ...

3, 5, 7, 9, 11, 13, 15, 17, 19, 21, ...

2

3, 4, 5, 6, 7, 8, 9, 10, ... 3, 5, 7, 11, 13, 17, 19, ...

2, 3, 5, 7, 11, 13, 17, 19, ...

Korutyny i wπtki
Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

czytaj([H | T], H, T).

pisz(H, [H | T], T).

zamknij([]).

Korutyny i wπtki
Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

sito(In, Out) :-
freeze(In,

( czytaj(In, N, In_)
-> pisz(N, Out, Out_),

filtruj(N, In_, Out1),
sito(Out1, Out_)

; zamknij(Out))).

Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

filtruj(N, In, Out) :-
freeze(In,

( czytaj(In, I, In_)
-> ( I mod N =:= 0

-> filtruj(N, In_, Out)
; pisz(I, Out, Out_),

filtruj(N, In_, Out_))
; zamknij(Out))).

Korutyny przekazujπce sobie dane w strumieniach

Example (cd.)

?- sito(S1, S2), generowanie(2, 20, S1).
S2 = [2, 3, 5, 7, 11, 13, 17, 19],
S1 = [2, 3, 4, 5, 6, 7, 8, 9, 10|...].

Tworzenie wπtku

I Do tworzenia wπtków s≥uøy predykat thread_create(Goal,
ID, Options) .

I Nowoutworzony wπtek wykonuje zadany cel Goal.
I Identyfikator utworzonego wπtku zwracany jest drugim

parametrem ID.
I Tworzeniem wπtku moøna sterowaÊ podajπc na liúcie Options

termy ustalajπce parametry.

Tworzenie wπtku

thread_create

thread_join

thread_create

thread_exit

detached(false)
domyślne

detached(true)

thread_exit

thread_exit

Tworzenie wπtku

Example
Utworzony wπtek nie wspó≥dzieli zmiennych z wπtkiem, który go
utworzy≥.

ex(Status) :-
thread_create(X = b, ID, []),
X = a,
thread_join(ID, Status).

Jak widaÊ w obu wπtkach pod zmiennπ X podstawiane sπ dwie
róøne (nieunifikowalne) wartoúci:

?- ex(S).
S = true.

Tworzenie wπtku

Wybrane opcje predykatu thread_create/3:
detached(Bool) Jeúli false (domyúlnie), to trzeba po≥πczyÊ siÍ z

tym wπtkiem po jego zakoÒczeniu wywo≥ujπc
predykat thread_join/2 (zostanπ swolnione
wszystkie jego zasoby). Gdy true, to automatycznie
po jego zakoÒczeniu zostanπ zwolnione wszystkie
jego zasoby.

queue_max_size(Size) Ustalenie maksymalnego rozmiaru kolejki
komunikatów.

Po≥πczenie wπtku

I Wywo≥anie predykatu thread_join(+Id, -Status) czeka
na zakoÒczenie wπtku Id i oddaje Status jego zakoÒczenia
(nastÍpujÍ po≥πczenie). Po zakoÒczeniu wπtku Id wszystkie
jego zasoby zostajπ zwolnione. Wπtek z atrybutem
detached(true) nie moøe byÊ po≥πczony,

Komunikacja miÍdzy wπtkami

I Predykat thread_send_message(+TID, +Term) wysy≥a do
wπtku TID komunikat w postaci termu Term. Po wys≥aniu
wπtek wysy≥ajπcy kontynuuje pracÍ a komunikat zostaje
dopisany na koÒcu kolejki komunikatów wπtku TID.

I Predykat thread_get_message(?Pattern) pobiera z kolejki
komunikatów ten, który unifikuje siÍ z zadanym wzorcem
Pattern i usuwa go z kolejki (komunikaty nieunifikujπce siÍ
pozostajπ w kolejce). Jeúli kolejka jest pusta, to dzia≥anie
wπtku jest wstrzymane aø do chwili pojawienia siÍ w kolejce
nowego komunikatu.

I Predykat thread_peek_message(?Pattern) dzia≥a
podobnie do thread_get_message/1 ale z tπ róønicπ, øe w
przypadku pustej kolejki komunikatów zawodzi (nie
wstrzymuje wπtku).

Komunikacja miÍdzy wπtkami

Example (Ping-pong)

main :-
thread_create(gracz, Id1, [detached(true)]),
thread_create(gracz, Id2, [detached(true)]),
thread_send_message(Id1, przeciwnik(Id2)),
thread_send_message(Id2, przeciwnik(Id1)),
thread_send_message(Id1, ping).

Komunikacja miÍdzy wπtkami

Example (Ping-pong cd.)

gracz :-
thread_get_message(przeciwnik(Id)),
gracz(Id).

gracz(Id) :-
thread_get_message(M1),
writeln(M1),
odbicie(M1, M2),
thread_send_message(Id, M2),
gracz(Id).

odbicie(ping, pong).
odbicie(pong, ping).

Komunikacja miÍdzy wπtkami

Example (Ping-pong cd.)
Efekt dzia≥ania:

?- main.
ping
pong
ping
pong
...

Synchronizacja wπtków

I Do synchronizacji wπtków s≥uøπ MUTEXy (ang. MUTual
EXclusion).

I Jeúli jakiú zasób jest wspó≥dzielony przez dwa lub wiÍcej
wπtków, to fragment kodu wymagajπcy posiadania tego
zasobu na wy≥πcznoúÊ nazywa siÍ sekcjπ krytycznπ.

I Przed wejúciem do sekcji krytycznej naleøy zablokowaÊ (ang.
lock) mutex zwiπzany z zasobem a po wyjúciu z sekcji
krytycznej naleøy mutex odblokowaÊ (ang. unlock).

I Jeúli mutex jest zablokowany, to kolejny wπtek, który chce go
zablokowaÊ przed wejúciem do sekcji krytycznej, zostanie
wstrzymany do chwili gdy mutex zostanie odblokowany przez
wπtek opuszczajπcy sekcjÍ krytycznπ.

I DziÍki mutexom mamy zagwarantowane, øe w sekcji
krytycznej zwiπzanej z danym zasobem znajduje siÍ co
najwyøej jeden wπtek.

Synchronizacja wπtków

I Predykat mutex_create(?Id) tworzy mutex Id.
I Predykat mutex_lock(+Id) albo zablokuje mutex Id i

skoÒczy siÍ powodzeniem albo wstrzyma dzia≥anie wπtku, jeúli
mutex Id jest zablokowany.

I Predykat mutex_trylock(+Id) dzia≥a podobnie jak
mutex_lock(+Id) ale gdy mutex jest zablokowany, to koÒczy
siÍ niepowodzeniem zamiast wstrzymywaÊ wπtek.

I Predykat mutex_unlock(+Id) odblokowuje mutex Id.
I Predykat with_mutex(+Id, Goal) blokuje mutex Id na czas

wykonywania celu Goal. Bez wzglÍdu na wynik wykonania
celu, zostajπ usuniÍte moøliwe punktu nawrotu oraz
odblokowywany jest mutex.

I Predykat mutex_destroy(+Id) usuwa mutex Id.

Synchronizacja wπtków

Example (B≥Ídne odliczanie)

main1 :-
thread_create(powtarzaj, _, [detached(true)]),
thread_create(powtarzaj, _, [detached(true)]).

powtarzaj :-
odliczaj(9), nl,
powtarzaj.

odliczaj(0).
odliczaj(N) :-

N > 0,
write(N),
N1 is N-1,
odliczaj(N1).

Synchronizacja wπtków

Example (B≥Ídne odliczanie cd.)
Przyk≥ad uruchomienia:

?- main1.
987654321
987654321
9876543219876543
9876521
987654321
9874321
987654654321
987654321
9876321
9876543543221
...

Synchronizacja wπtków

Example (Poprawne odliczanie)

main2 :-
mutex_create(Mutex),
thread_create(powtarzaj(Mutex), _, [detached(true)]),
thread_create(powtarzaj(Mutex), _, [detached(true)]).

powtarzaj(Mutex) :-
with_mutex(Mutex, (odliczaj(9), nl)),
powtarzaj(Mutex).

Synchronizacja wπtków

Example (Poprawne odliczanie cd.)
Przyk≥ad uruchomienia:

?- main2.
987654321
987654321
987654321
987654321
987654321
987654321
987654321
987654321
987654321
987654321
987654321
...


